

TD

Méta-analyse sur l'efficacité  
d'injection de Mg sur le risque  
d'infarctus

# 14 études sur les taux de mortalité avec/sans Mg

(Sterne et al., 2001)


Chaque étude a

- 1 groupe traité
- 1 groupe témoin

Groupe traité avec du Mg

Groupe témoin

Nombre total de patients



## Odds et Odds ratio

$$Odds = \frac{Proba\ de\ mortalité}{1 - Proba\ de\ mortalité}$$

$$Odds\ ratio = \frac{Odds(avec\ Mg)}{Odds(sans\ Mg)}$$

## Odds et Odds ratio

$$Odds = \frac{Proba\ de\ mortalité}{1 - Proba\ de\ mortalité}$$

$$Odds\ ratio = \frac{Odds(avec\ Mg)}{Odds(sans\ Mg)}$$

Si le *Odds ratio* est  $>1$  (ou  $\ln Odds\ ratio > 0$ ),  
le traitement n'est pas efficace

## Objectif du TD

Estimer le Odds ratio à l'aide des 14 études en tenant compte de l'hétérogénéité des études

Estimer la probabilité que le Odds ratio soit  $>1$  en moyenne dans la population et pour une nouvelle étude

# Un modèle hiérarchique (Higgins et Spiegelhalter, 2002)

$$r_{ci} \sim \text{Bin}(n_{ci}, p_{ci})$$

$$r_{mi} \sim \text{Bin}(n_{mi}, p_{mi})$$

$$\varphi_i = \text{logit}(p_{ci}) = \ln \left( \frac{p_{ci}}{1 - p_{ci}} \right)$$

$$\text{logit}(p_{mi}) = \ln \left( \frac{p_{mi}}{1 - p_{mi}} \right) = \varphi_i + \delta_i$$

$$\delta_i \sim N(\mu, \tau^2)$$

$$r_{ci} \sim \text{Bin}(n_{ci}, p_{ci})$$

$$r_{mi} \sim \text{Bin}(n_{mi}, p_{mi})$$

$$\varphi_i = \text{logit}(p_{ci}) = \ln \left( \frac{p_{ci}}{1 - p_{ci}} \right)$$

$$\text{logit}(p_{mi}) = \ln \left( \frac{p_{mi}}{1 - p_{mi}} \right) = \varphi_i + \delta_i$$

$$\delta_i \sim N(\mu, \tau^2)$$

**$\delta_i = \ln \text{Odds ratio } i$**

**$\mu = \ln \text{Odds ratio population}$**

$$r_{ci} \sim \text{Bin}(n_{ci}, p_{ci})$$

$$r_{mi} \sim \text{Bin}(n_{mi}, p_{mi})$$

$$\varphi_i = \text{logit}(p_{ci}) = \ln\left(\frac{p_{ci}}{1 - p_{ci}}\right)$$

$$\text{logit}(p_{mi}) = \ln\left(\frac{p_{mi}}{1 - p_{mi}}\right) = \varphi_i + \delta_i$$

$$\delta_i \sim N(\mu, \tau^2)$$

## **PRIOR**

$$p_{ci} \sim \text{Unif}(0, 1)$$

$$\tau \sim \text{Unif}(0, 100)$$

$$\mu \sim N(0, 10^4)$$



# WinBUGS

```
list(K=14,  
  
#Nb de morts dans le groupe avec Mg  
rm= c(1,9,2,1,10,1,1,90,0,6,1,5,4,4),  
#Nb de morts dans le groupe sans Mg  
rc=c(2,23,7,1,8,9,3,118,1,11,7,13,8,17),  
  
#Effectif dans le groupe avec Mg  
nm=c(40,135,200,48,150,59,25,1159,22,76,27,23,130,107),  
#Effectif dans le groupe sans Mg  
nc=c(36,135,200,46,148,56,23,1157,21,75,27,33,122,108)  
)
```

```

model {

for(i in 1:K) {

#Lois binomiales
rc[i] ~ dbin(pc[i],nc[i]);
rm[i] ~ dbin(pm[i],nm[i]);

#Logit
phi[i] <- logit(pc[i]);
logit(pm[i]) <- phi[i] + delta[i];

#log(odds ratio)
delta[i] ~ dnorm(mu, precision);

#Odds ratio pour l'étude i
Odds[i] <-exp(delta[i]);

#Prior pour pc
pc[i] ~ dunif(0,1)
}

```

```

#Odds ratio nlle etude
delta.new ~ dnorm(mu, precision);
Odds.new<-exp(delta.new);

#Odds ratio population
Odds.pop<-exp(mu);


#prior pour le log Odds ratio moyen
mu ~ dnorm(0.0, 0.0001);

#Prior pour tau
tau ~ dunif(0, 100);
precision <- 1/(tau*tau);
tau.sq <- 1/precision;


#Probabilite log(Odds ratio)<0
less0 <- min(delta.new, 0);
less0mean<-min(mu, 0);
prob0 <- 1 - equals(less0, 0);
prob0mean<- 1 - equals(less0mean, 0);
}

```

## Odds de la population d'études





## Odds des études 1, 2 et 3




# Odds ratios des études individuelles

node	mean	sd	MC error	2.5%	median	97.5%	start	sample
Odds[1]	0.4066	0.2536	0.002132	0.09683	0.3602	1.012	1000	39000
Odds[2]	0.367	0.1247	0.001303	0.1701	0.3519	0.6498	1000	39000
Odds[3]	0.3503	0.1704	0.001767	0.1018	0.3232	0.7505	1000	39000
Odds[4]	0.4479	0.3007	0.001841	0.1059	0.3881	1.182	1000	39000
Odds[5]	0.7502	0.3452	0.003878	0.3283	0.6679	1.633	1000	39000
Odds[6]	0.2689	0.1443	0.002013	0.05795	0.2463	0.6039	1000	39000
Odds[7]	0.374	0.2199	0.001905	0.08591	0.337	0.8838	1000	39000
Odds[8]	0.6977	0.1034	9.54E-4	0.5157	0.6906	0.9207	1000	39000
Odds[9]	0.3652	0.2422	0.002131	0.0609	0.3232	0.9312	1000	39000
Odds[10]	0.4463	0.1855	0.001306	0.179	0.4159	0.8938	1000	39000
Odds[11]	0.2931	0.1585	0.001912	0.06571	0.2683	0.663	1000	39000
Odds[12]	0.4223	0.191	0.00146	0.1548	0.3897	0.8851	1000	39000
Odds[13]	0.4184	0.1874	0.001459	0.1522	0.3889	0.8738	1000	39000
Odds[14]	0.2938	0.1246	0.001683	0.1035	0.2762	0.5813	1000	39000

# Odds ratios des études individuelles et effet « shrinkage »




## Odds ratio moyen pour la population d'études



node	mean	sd	MC error	2.5%	median	97.5%	start	sample
Odds.pop	0.3726	0.1003	0.001522	0.196	0.3655	0.5877	1000	39000


## Odds ratio pour une nouvelle d'étude



node	mean	sd	MC error	2.5%	median	97.5%	start	sample
Odds.new	0.4673	0.7198	0.003881	0.07487	0.3757	1.388	1000	39000



## Odds ratio pour une nouvelle d'étude



Plus de 5% de chance d'avoir une nouvelle étude avec un Odds ratio >1

node	mean	sd	MC error	2.5%	median	97.5%	start	sample
Odds.new	0.4673	0.7198	0.003881	0.07487	0.3757	1.388	1000	39000

# Question supplémentaire : Refaire l'analyse avec un prior plus informatif

A priori « sceptique »

5% de chance que le odds ratio soit inférieur à 0.75

$$\mu \sim N(0, 0.03)$$